初級講座

プログラミングで学ぶ有限要素法
4. 弾塑性モデルによる応力–ひずみ関係の計算

山 川 優 樹（やまかわ ゆうき）
東北大学大学院准教授 工学研究科

宮 田 喜 壽（みやた よしひさ）
防衛大学校准教授 システム工学群

渦 岡 良 介（うずおか りょうすけ）
徳島大学大学院教授 ソシオテクノサイエンス研究部

4.1 土の材料特性と弾塑性モデル

3章で述べた有限要素法による一次元変形解析（全応力解析）では，材料モデルに線形弾性モデルを用いた。しかしながら，実際の土の応力–ひずみ関係[注1)]は複雑であり，その正確な計算は地盤工学の数値解析において重要である[1)]。そこで，本章では弾塑性の応力–ひずみ関係とその計算法を解説する。本章の前半では，弾塑性モデルへの入門として，土の等方圧縮特性を表す際によく用いられる「e–log σ 関係[注2)]」に基づく一次元弾塑性モデルを考える。本章の後半では，弾塑性モデルの応力計算法を詳しく解説する。なお，弾塑性モデルを有限要素法に組み込んだ変形解析や圧密解析については，次号以降に掲載予定の5章と6章で解説する[注3)]。

実務における地盤解析では，最近では三次元解析も多く行われている。その際には，本章で扱う一次元モデルよりも一般的な三次元応力状態を考慮した構成モデルが用いられる。よく用いられるモデルの代表例として，カムクレイモデル[2),3)]が挙げられる。このような弾塑性モデルを難しく感じている読者も多いかも知れないが，実はカムクレイモデルの塑性硬化・軟化特性は e–log σ 関係を基礎とした圧密降伏応力と塑性体積ひずみとの関係によって表される。このことから，本章で一次元モデルの基本構成とその計算法をしっかり習得すれば，より一般性のある三次元モデルの理解も容易になるはずである。

4.2 一次元圧縮を対象とした弾塑性モデル

4.2.1 e–ln σ 関係の整理

最初に e–ln σ 関係[注4)]について整理しておく。簡単のため，引張りには抵抗しない地盤材料を想定し，応力 σ や圧密降伏応力 σ_c は圧縮側（$\sigma<0$, $\sigma_c<0$）のみとし，これらと間隙比 e との関係を考える。

練り返した粘土を圧縮したときの典型的な挙動を e–ln σ 関係で表すと，図—4.1のようになる。図中のAは応力 σ_0，間隙比 e_0 の基準状態（以下ではA（σ_0, e_0）と表す）である。Aから圧縮応力を大きくしていくと，土は圧縮して（間隙比は小さくなって），Bを経てC（σ_c, e_c）に達する。この過程では，圧縮応力の増加に対する圧縮変形の進行（間隙比の減少）の程度がある時点から大きくなっており，この点をBとしている。次に，Cから圧縮応力を σ_0 まで小さくすると，間隙比は e_0 までは戻らずに e_p となり，D（σ_0, e_p）に達する。すなわち，除荷するとC→B→AではなくC→Dをたどり，間隙比の差 e_p-e_0 の分だけ永久変形が生じている。Dから圧縮応力を再び σ_c まで大きくすると，C→Dとほぼ同じ経路を逆にたどってEに達する。Eでの間隙比はCでの値 e_c とほぼ同じである。以上のように，C→D, D→Eは可逆的・弾性的である。A→Bも同様である。一方，B→Cでは永久変形が発生しており，弾塑性的な挙動である。

以上で述べた図—4.1の挙動について，A, B, C, D, Eの各点間をそれぞれ直線とみなしてモデル化すると，図—4.2のようになる。ここで，弾性的な挙動であるA→B, C→D, D→E間の直線の傾きを κ とし，C→D→E間は可逆的であるとした。また，弾塑性的な挙動であるB→C間の傾きを λ とした。この変形係数 λ, κ はそれぞれ圧縮指数・膨潤指数と呼ばれる[注5)]。なお，直線BC上の状態を正規圧密状態と呼び，直線BCは正規圧密線



図—4.1 粘土を圧縮したときの典型的な e–ln σ 関係

注1) 5章の圧密解析（有効応力解析）では，土骨格の有効応力–ひずみ関係に相当する。

注2) この関係は一般に「e–log p' 関係」（ここで p' は圧力から間隙水圧を差し引いた有効圧力）と呼ぶことが多い。しかし，ここでは一次元モデルを考えており，応力を σ と表記しているため，「e–log σ 関係」と呼ぶ。なお，本講座では一貫して応力・ひずみは引張り（膨張）を正と定義しているため，一連の式やグラフでは log (−σ) となっていることに注意されたい。

注3) 6章の圧密解析で土骨格の有効応力–ひずみ関係を考える際には，全応力から間隙水圧を差し引いた有効応力 σ' を用いるので，本章で解説する弾塑性モデル諸式の σ を σ' に置き換えればよい。

注4) ここでは自然対数を用いた e–ln σ 関係を考えるが，常用対数もしばしば用いられる。詳しくは後述する。



図―4.2 実挙動をモデル化した e–ln σ 関係

と呼ぶ。また，直線BCよりも下側（左側）の状態を過圧密状態と呼び，線分AB, CD, DEを膨潤線・再圧密線などと呼ぶ[注6]。このとき，弾性領域について

$$e=e_{\mathrm{p}}-\kappa\ln\frac{\sigma}{\sigma_0}\quad\cdots\cdots(4.1)$$

の関係が成り立つ。また，正規圧密の挙動である弾塑性領域について

$$e_{\mathrm{c}}=e_{\mathrm{c0}}-\lambda\ln\frac{\sigma_{\mathrm{c}}}{\sigma_{\mathrm{c0}}}\quad\cdots\cdots(4.2)$$

の関係が成り立つ[注7]。**図―4.2**を参照すると，

$$e_{\mathrm{c}}-e_{\mathrm{c0}}=e_{\mathrm{p}}-e_0-\kappa\ln\frac{\sigma_{\mathrm{c}}}{\sigma_{\mathrm{c0}}}\quad\cdots\cdots(4.3)$$

の関係が得られ，さらに式(4.2), (4.3)より次式が導かれる。

$$e_{\mathrm{p}}-e_0=-(\lambda-\kappa)\ln\frac{\sigma_{\mathrm{c}}}{\sigma_{\mathrm{c0}}}\quad\cdots\cdots(4.4)$$

4.2.4で弾塑性の応力–ひずみ関係を導出するための準備として，式(4.1), (4.4)を増分形[注8]で表しておく。各式の微分を取ると，それぞれ

$$\mathrm{d}e=-(\lambda-\kappa)\frac{\mathrm{d}\sigma_{\mathrm{c}}}{\sigma_{\mathrm{c}}}-\kappa\frac{\mathrm{d}\sigma}{\sigma},\quad\cdots\cdots(4.5)$$

$$\mathrm{d}e_{\mathrm{p}}=-(\lambda-\kappa)\frac{\mathrm{d}\sigma_{\mathrm{c}}}{\sigma_{\mathrm{c}}}\quad\cdots\cdots(4.6)$$

となる。上の2式で，'d'が付いた諸量は微小増分である。また，基準状態における諸量 e_0, σ_0, σ_{c0} は変形によらず一定であり，それらの微分はいずれもゼロであることを用いている。

4.2.2 間隙比の変化とひずみの定義

4.2.1で整理した e–ln σ 関係に基づいて弾塑性の応力–ひずみ関係を導く準備として，土の変形の尺度としてのひずみを定義する。

注5) それぞれ，弾塑性圧縮指数・弾性圧縮指数と呼ばれることもある。

注6) ここでは過圧密状態の挙動を完全な弾性とモデル化して膨潤線と再圧密線を同一視しているが，実際の土の挙動は**図―4.1**のようにヒステリシスループを描くことが多い。

注7) 常用対数を用いた e–$\log_{10}\sigma$ 関係における圧縮指数・膨潤指数（それぞれ C_{c}, C_{s} という記号で表すことが多い。添字の'c'と's'は'compression'と'swelling'の意味である）は，それぞれ式(4.1), (4.2)における圧縮指数 λ，膨潤指数 κ の約2.3倍の値となる（対数の底の変換と $\ln 10\approx 2.3$ の関係から導かれる）。すなわち，$1/\ln 10\approx 0.434$ より，$\lambda=0.434\,C_{\mathrm{c}}$, $\kappa=0.434\,C_{\mathrm{s}}$ の関係がある。

注8) 本講座では初学者に馴染みやすいように「増分形」を考えるが，一般的な弾塑性力学では時間微分を考えて「速度形」を用いることが多い。

まず，間隙比 e と土の変形（体積変化）との関係を整理する。本講座では固相（土粒子）と液相（間隙水）の2相からなる飽和土を考えており，土粒子および間隙水は非圧縮（体積変化しない）と仮定している。この仮定のもとでは，土全体としての変形（圧縮・膨張）は間隙水の出入りによって生じることになる。ここで，土の体積を V とし，そのうち土粒子が占める体積を V_{s}，間隙が占める体積を V_{v} とすると，$V=V_{\mathrm{s}}+V_{\mathrm{v}}$ の関係がある。土の間隙がどの程度かを表す指標としてよく用いられる間隙比 e は，次のように定義される。

$$e=V_{\mathrm{v}}/V_{\mathrm{s}}\quad\cdots\cdots(4.7)$$

また，次式で定義される比体積 v もよく用いられる。

$$v=\frac{V}{V_{\mathrm{s}}}=\frac{V_{\mathrm{s}}+V_{\mathrm{v}}}{V_{\mathrm{s}}}=1+\frac{V_{\mathrm{v}}}{V_{\mathrm{s}}}=1+e\quad\cdots\cdots(4.8)$$

式(4.8)より，**図―4.2**における基準状態A（σ_0, e_0）ならびに変形後の状態（σ, e）での体積は，比体積 v と固相の体積 V_{s} を用いて，それぞれ次のように表される。

$$\text{基準状態での体積：}V_0=v_0V_{\mathrm{s0}}$$

$$\text{変形後の体積：}V=vV_{\mathrm{s}}\approx vV_{\mathrm{s0}}\quad\cdots\cdots(4.9)$$

上式で，各変数についている添字'0'は基準状態（図―4.2の点 A）での値であることを示す。また，前述のように土粒子は非圧縮と仮定しているため，$V_{\mathrm{s}}\approx V_{\mathrm{s0}}$ としている。

次に，ひずみを定義する。膨張変形を正，圧縮変形を負とし，微小ひずみの定義に従って，基準状態の体積 V_0 に対する体積変化量 $\varDelta V=V-V_0$ の割合をひずみ ε と定義すると，式(4.9)から次式のようになる。

$$\varepsilon=\frac{\varDelta V}{V_0}=\frac{V-V_0}{V_0}=\frac{vV_{\mathrm{s0}}-v_0V_{\mathrm{s0}}}{v_0V_{\mathrm{s0}}}=\frac{v-v_0}{v_0}=\frac{e-e_0}{1+e_0}\quad\cdots\cdots(4.10)$$

ひずみ ε は，次式のように弾性ひずみ ε^{e} と塑性ひずみ ε^{p} とに足し算の形で分解される（加算分解）。

$$\varepsilon=\varepsilon^{\mathrm{e}}+\varepsilon^{\mathrm{p}}\quad\cdots\cdots(4.11)$$

図―4.2を参照して，弾性ひずみ ε^{e} および塑性ひずみ ε^{p} はそれぞれ次のように表される。

$$\varepsilon^{\mathrm{e}}=\frac{e-e_{\mathrm{p}}}{1+e_0},\quad\varepsilon^{\mathrm{p}}=\frac{e_{\mathrm{p}}-e_0}{1+e_0}\quad\cdots\cdots(4.12)$$

4.2.4で応力–ひずみ関係を導くが，弾塑性の場合は一般に応力増分 $\mathrm{d}\sigma$ とひずみ増分 $\mathrm{d}\varepsilon$ との関係（増分形の応力–ひずみ関係）を用いる。その準備として，式(4.10), (4.12)で定義したひずみとその弾性・塑性部分を増分形で表すと，

$$\mathrm{d}\varepsilon=\frac{\mathrm{d}e}{1+e_0},\ \mathrm{d}\varepsilon^{\mathrm{e}}=\frac{\mathrm{d}e-\mathrm{d}e_{\mathrm{p}}}{1+e_0},\ \mathrm{d}\varepsilon^{\mathrm{p}}=\frac{\mathrm{d}e_{\mathrm{p}}}{1+e_0}\quad\cdots\cdots(4.13)$$

となる。ここでも，先の式(4.5), (4.6)と同様に，$\mathrm{d}e_0=0$ の関係を用いた。式(4.13)から明らかに，

$$\mathrm{d}\varepsilon=\mathrm{d}\varepsilon^{\mathrm{e}}+\mathrm{d}\varepsilon^{\mathrm{p}}\quad\cdots\cdots(4.14)$$

であり，式(4.11)に示したひずみの加算分解と同様に，ひずみ増分も弾性・塑性部分に加算分解される。

4.2.3 応力–ひずみ関係の導出

4.2.1で述べた e–ln σ 関係と，**4.2.2**でのひずみの定義に基づいて，応力–ひずみ関係を導く。具体的には，

弾性挙動に関する応力σ–弾性ひずみε^e関係（弾性構成式）と，塑性挙動に関する圧密降伏応力σ_c–塑性ひずみε^p関係（塑性硬化則）を導く。

まず，弾性構成式を導く。式(4.1)と式(4.12)第一式より，

$$\varepsilon^e = -\frac{\kappa}{1+e_0}\ln\frac{\sigma}{\sigma_0} \quad \cdots\cdots (4.15)$$

であり，これを式変形して弾性構成式が次のように導かれる。

$$\sigma = \sigma_0 \exp\left[-\frac{1+e_0}{\kappa}\varepsilon^e\right] \quad \cdots\cdots (4.16)$$

この弾性構成式の基本的な特性は以下のとおりである：

- 基準状態$\varepsilon^e=0$のとき，$\sigma=\sigma_0$,
- 大きな弾性圧縮$\varepsilon^e\to-\infty$のとき，$\sigma\to-\infty$,
- 大きな弾性膨張$\varepsilon^e\to+\infty$のとき，$\sigma\to 0$.

次項で必要となる増分形の弾性構成式は，式(4.16)を微分することにより，次のように導かれる。

$$\mathrm{d}\sigma = \frac{\mathrm{d}\sigma}{\mathrm{d}\varepsilon^e}\mathrm{d}\varepsilon^e = \underbrace{\sigma_0 \exp\left[-\frac{1+e_0}{\kappa}\varepsilon^e\right]}_{=\sigma}\left(-\frac{1+e_0}{\kappa}\right)\mathrm{d}\varepsilon^e$$

$$= -\frac{1+e_0}{\kappa}\sigma\mathrm{d}\varepsilon^e = D^e\mathrm{d}\varepsilon^e \quad \cdots\cdots (4.17)$$

ここで，弾性接線係数D^eを次のように定義している。

$$D^e = -\frac{1+e_0}{\kappa}\sigma \quad \cdots\cdots (4.18)$$

この式から分かるように，D^eは応力σによって変化する。例えば，圧縮応力が大きく（σ(<0)が小さく）なるほどD^eは大きくなる。つまり，e–$\ln\sigma$関係に基づく弾性構成式は非線形弾性となる。

次に，塑性硬化則を導く。式(4.4)と式(4.12)第二式より，

$$\varepsilon^p = -\frac{\lambda-\kappa}{1+e_0}\ln\frac{\sigma_c}{\sigma_{c0}} \quad \cdots\cdots (4.19)$$

であり，これより塑性硬化則が次のように導かれる。

$$\sigma_c = \sigma_{c0}\exp\left[-\frac{1+e_0}{\lambda-\kappa}\varepsilon^p\right] \quad \cdots\cdots (4.20)$$

この塑性硬化則の基本的な挙動を簡単に示しておく：

- 基準状態$\varepsilon^p=0$のとき，$\sigma_c=\sigma_{c0}$,
- 大きな塑性圧縮$\varepsilon^p\to-\infty$のとき，$\sigma_c\to-\infty$,
- 大きな塑性膨張$\varepsilon^p\to+\infty$のとき，$\sigma_c\to 0$.

式(4.20)を微分すれば，増分形の塑性硬化則

$$\mathrm{d}\sigma_c = \frac{\mathrm{d}\sigma_c}{\mathrm{d}\varepsilon^p}\mathrm{d}\varepsilon^p = \underbrace{\sigma_{c0}\exp\left[-\frac{1+e_0}{\lambda-\kappa}\varepsilon^p\right]}_{=\sigma_c}\left(-\frac{1+e_0}{\lambda-\kappa}\right)\mathrm{d}\varepsilon^p$$

$$= -\frac{1+e_0}{\lambda-\kappa}\sigma_c\mathrm{d}\varepsilon^p = K^p\mathrm{d}\varepsilon^p \quad \cdots\cdots (4.21)$$

が導かれる。ここで，塑性硬化係数K^pを

$$K^p = -\frac{1+e_0}{\lambda-\kappa}\sigma_c \quad \cdots\cdots (4.22)$$

と定義した。e–$\ln\sigma$関係から導かれる塑性硬化係数K^pは定数ではなく，σ_cによって変化する。例えば，圧密降伏応力が大きく（σ_c(<0)が小さく）なるほどK^pは大きくなる。

4.2.4 塑性変形の発生と進展，および弾性・塑性応答の判別条件に関する定式化

e–$\ln\sigma$関係に基づく弾塑性モデルを完成させるためには，既述のひずみの加算分解および弾性構成式・塑性硬化則に加えて，(i)どのようなときに塑性変形が生じるのか，(ii)塑性変形はどのように進展するのかの2点を規定する必要がある。

まず，(i)については，**図―4.2**からも分かるように，弾性・弾塑性いずれの状態においても，ある間隙比eに対して応力σは圧密降伏応力σ_cを越えることはない。σおよびσ_cともに圧縮応力なので負値であることを考慮すると，(i)は次式により規定できる。

$$f(\sigma,\sigma_c) = |\sigma| - |\sigma_c| = \sigma_c - \sigma \le 0 \quad \cdots\cdots (4.23)$$

この関数$f(\sigma,\sigma_c)$は弾塑性力学の分野で「降伏関数」と呼ばれる。$f=0$の状態は現在状態(σ, e)が正規圧密線上に位置すること（正規圧密状態）に相当し，その直後の応答は，塑性負荷（正規圧密線をたどる）と弾性除荷（膨潤線をたどって過圧密領域へいく）の2とおりが考えられる。前者の塑性負荷が継続する過程では，常に$f=0$，すなわち$\sigma=\sigma_c$の関係が成り立つ（σが変化すると，それに応じてσ_cも変化する）ことに着目されたい。一方，$f<0$の状態は，現在状態(σ, e)が正規圧密線の下側に位置すること（過圧密状態）に相当し，その直後の応答は，膨潤線をたどって更に除荷が進行するか，再載荷となるかが考えられるが，いずれにせよ弾性応答となる。なお，前述のとおり，$f>0$（すなわち$|\sigma|>|\sigma_c|$）となることは無い。つまり，点(σ, e)が正規圧密線よりも上側に位置するような状態は，通常の弾塑性モデルでは許容されない。

次に，(ii)については，塑性ひずみの微小増分$\mathrm{d}\varepsilon^p$をその「大きさ」と「方向」に分けて規定することにしよう。まず，「大きさ」を非負の係数$\mathrm{d}\gamma(\ge 0)$で表す。次に「方向」は，現在の応力状態に応じて決まるものとする。ここでは一次元モデルを考えているので，塑性ひずみの進展方向は引張りまたは圧縮の2方向だけであり，更にここでは圧縮応力（$\sigma<0$）だけを考えるので，塑性ひずみは圧縮側にのみ進展することになる。式(4.23)を考慮しつつ，以上の議論を式で表すと，

$$\mathrm{d}\varepsilon^p = \mathrm{d}\gamma\frac{\partial f}{\partial\sigma} = \mathrm{d}\gamma\,\mathrm{sign}(\sigma) = -\mathrm{d}\gamma \quad \cdots\cdots (4.24)$$

となる。上式は塑性ひずみの進展を規定する式であり，弾塑性力学の分野では「塑性流れ則」と呼ばれる[注9)]。式中の$\mathrm{sign}(\sigma)$はσの正負に応じて±1の値をとるが，ここでは常に$\sigma<0$なので$\mathrm{sign}(\sigma)=-1$となる。また，係数$\mathrm{d}\gamma(\ge 0)$は「塑性乗数」などと呼ばれる。

以上の議論から，材料の応答が弾性であるか，弾塑性

注9) 式(4.24)で塑性ひずみの進展方向を$\frac{\partial f}{\partial\sigma}$としているのを疑問に感じた読者もいるかと思うが，これは一般的な三次元の弾塑性モデルにおける塑性流れ則に即した形式で表したためである。このような形の塑性流れ則は「関連流れ則」と呼ばれる。三次元の弾塑性モデルに興味のある読者は，専門書[3),4)]を参照されたい。

であるかについて，次の条件が導かれる。

$$f<0 \Longrightarrow \mathrm{d}\gamma=0 \quad (\text{弾性応答})$$
$$\mathrm{d}\gamma>0 \Longrightarrow f=0 \quad (\text{弾塑性応答}) \cdots\cdots(4.25)$$

上式より，弾性・弾塑性のいずれの応答にせよ，次の条件式が常に成り立ち，「負荷・除荷条件」と呼ばれる。

$$\mathrm{d}\gamma\geq 0,\ f(\sigma, \sigma_c)\leq 0,\ \mathrm{d}\gamma\cdot f(\sigma, \sigma_c)=0 \cdots\cdots(4.26)$$

最後に，$f=0$ の状態に続く応答について考える。先に述べたとおり，$f=0$ の状態に続く応答は塑性負荷と弾性除荷の2とおりが考えられる。そのうち，塑性負荷が持続する場合は常に $f=0$ であるから，$\mathrm{d}f=0$（f の値がゼロのまま変化しないことを意味する）が成り立つ。一方，式(4.23)より $f\not>0$ であるから，$f=0$ の状態においては $\mathrm{d}f\not>0$ であり，弾性除荷の場合には $\mathrm{d}f<0$ となる。これを式で表すと，

$$\mathrm{d}f<0 \Longrightarrow \mathrm{d}\gamma=0 \quad (\text{弾性応答})$$
$$\mathrm{d}\gamma>0 \Longrightarrow \mathrm{d}f=0 \quad (\text{弾塑性応答}) \cdots\cdots(4.27)$$

上式より，$f=0$ の状態に対する後続の応答に関して，次の「塑性適合条件」が導かれる。

$$\mathrm{d}\gamma\cdot \mathrm{d}f(\sigma, \sigma_c)=0 \quad (f(\sigma, \sigma_c)=0 \text{ の状態について}) \cdots\cdots(4.28)$$

4.2.5　諸式の整理と弾塑性の増分形の応力-ひずみ関係の導出

以上で述べた諸式を整理すると，弾塑性モデルは以下の a)～g）により構成される。

a）ひずみの加算分解

$$\varepsilon=\varepsilon^e+\varepsilon^p \rightsquigarrow \text{増分形 } \mathrm{d}\varepsilon=\mathrm{d}\varepsilon^e+\mathrm{d}\varepsilon^p \cdots\cdots(4.29)$$

b）弾性構成式

$$\sigma=\sigma_0 \exp\left[-\frac{1+e_0}{\kappa}\varepsilon^e\right] \rightsquigarrow \text{増分形 } \mathrm{d}\sigma=D^e\mathrm{d}\varepsilon^e \cdots\cdots(4.30)$$

c）塑性硬化則

$$\sigma_c=\sigma_{c0} \exp\left[-\frac{1+e_0}{\lambda-\kappa}\varepsilon^p\right] \rightsquigarrow \text{増分形 } \mathrm{d}\sigma_c=K^p\mathrm{d}\varepsilon^p \cdots\cdots(4.31)$$

d）降伏関数

$$f(\sigma, \sigma_c)=\sigma_c-\sigma\leq 0 \cdots\cdots(4.32)$$

e）塑性流れ則

$$\mathrm{d}\varepsilon^p=\mathrm{d}\gamma\frac{\partial f}{\partial \sigma} \cdots\cdots(4.33)$$

f）負荷・除荷条件

$$\mathrm{d}\gamma\geq 0,\ f(\sigma, \sigma_c)\leq 0,\ \mathrm{d}\gamma\cdot f(\sigma, \sigma_c)=0 \cdots\cdots(4.34)$$

g）塑性適合条件

$$\mathrm{d}\gamma\cdot \mathrm{d}f(\sigma, \sigma_c)=0 \ (f(\sigma, \sigma_c)=0 \text{ の状態について}) \cdots\cdots(4.35)$$

上の a)～g）に基づいて，以下では弾塑性の応力-ひずみ関係を導出する。弾塑性モデルの場合には，式(4.30)の第一式の弾性構成式のように応力とひずみとの関係で表すことはできず，応力増分とひずみ増分との関係（増分形の応力-ひずみ関係）として表される。なぜなら，弾塑性材料では塑性負荷と弾性除荷の2とおりの応答が考えられ，材料が現在状態に至るまでに負荷・除荷の履歴を経験している場合，応力とひずみとを一対一に関係付けることはもはやできないからである。

まず，g）の塑性適合条件より，弾塑性応答時（$\mathrm{d}\gamma>0$）には $\mathrm{d}f(\sigma, \sigma_c)=0$ であり，式(4.29), (4.30), (4.31), (4.33)を用いると，次のようになる。

$$\begin{aligned}\mathrm{d}f(\sigma, \sigma_c)&=\frac{\partial f}{\partial \sigma}\mathrm{d}\sigma+\frac{\partial f}{\partial \sigma_c}\mathrm{d}\sigma_c\\&=\frac{\partial f}{\partial \sigma}D^e(\mathrm{d}\varepsilon-\mathrm{d}\varepsilon^p)+\frac{\partial f}{\partial \sigma_c}K^p\mathrm{d}\varepsilon^p\\&=\frac{\partial f}{\partial \sigma}D^e\left(\mathrm{d}\varepsilon-\mathrm{d}\gamma\frac{\partial f}{\partial \sigma}\right)+\frac{\partial f}{\partial \sigma_c}K^p\mathrm{d}\gamma\frac{\partial f}{\partial \sigma}\\&=0 \cdots\cdots(4.36)\end{aligned}$$

これを $\mathrm{d}\gamma$ について解くと，次のようになる。

$$\mathrm{d}\gamma=\frac{\dfrac{\partial f}{\partial \sigma}D^e\mathrm{d}\varepsilon}{\dfrac{\partial f}{\partial \sigma}D^e\dfrac{\partial f}{\partial \sigma}-\dfrac{\partial f}{\partial \sigma_c}K^p\dfrac{\partial f}{\partial \sigma}} \cdots\cdots(4.37)$$

式(4.30)第二式の増分形の弾性構成式に式(4.29), (4.33), (4.37)を代入すると，

$$\begin{aligned}\mathrm{d}\sigma&=D^e(\mathrm{d}\varepsilon-\mathrm{d}\varepsilon^p)\\&=D^e\left(\mathrm{d}\varepsilon-\mathrm{d}\gamma\frac{\partial f}{\partial \sigma}\right)\\&=\underbrace{\left[D^e-\frac{\left(D^e\dfrac{\partial f}{\partial \sigma}\right)\left(\dfrac{\partial f}{\partial \sigma}D^e\right)}{\dfrac{\partial f}{\partial \sigma}D^e\dfrac{\partial f}{\partial \sigma}-\dfrac{\partial f}{\partial \sigma_c}K^p\dfrac{\partial f}{\partial \sigma}}\right]}_{=D^{ep}}\mathrm{d}\varepsilon \cdots\cdots(4.38)\end{aligned}$$

となり，弾塑性の増分形の応力-ひずみ関係が次のように導出される。

$$\mathrm{d}\sigma=D^{ep}\mathrm{d}\varepsilon \cdots\cdots(4.39)$$

D^{ep} は弾塑性接線係数と呼ばれる[注10]。式(4.32)より $\dfrac{\partial f}{\partial \sigma}=-1,\ \dfrac{\partial f}{\partial \sigma_c}=1$ であることを用いると，D^{ep} は

$$D^{ep}=\begin{cases}D^e & \text{弾性応答（}\mathrm{d}\gamma=0\text{）のとき}\\ \dfrac{D^eK^p}{D^e+K^p} & \text{弾塑性応答（}\mathrm{d}\gamma>0\text{）のとき}\end{cases} \cdots\cdots(4.40)$$

となる。弾性接線係数 D^e の具体形は，既に式(4.18)に示した。弾塑性応答のときの D^{ep} の具体形は，式(4.18)の D^e および式(4.22)の K^p を式(4.40)へ代入することにより得られる。塑性負荷のとき，$f=0$ より $\sigma=\sigma_c$ であるから，

$$\begin{aligned}D^{ep}&=\frac{D^eK^p}{D^e+K^p}=\frac{\left(-\dfrac{1+e_0}{\kappa}\sigma_c\right)\cdot\left(-\dfrac{1+e_0}{\lambda-\kappa}\sigma_c\right)}{\left(-\dfrac{1+e_0}{\kappa}\sigma_c\right)+\left(-\dfrac{1+e_0}{\lambda-\kappa}\sigma_c\right)}\\&=-\frac{1+e_0}{\lambda}\sigma_c \cdots\cdots(4.41)\end{aligned}$$

となる。以上より，e–$\ln\sigma$ 関係に基づく弾塑性モデルの D^{ep} は次のように与えられる。

注10）式(4.38)中の D^{ep} が複雑な形で表されていることを疑問に思う読者がいるかも知れないが，多次元の弾塑性モデルの弾塑性接線係数と対比しやすいよう，あえてこのような形で表している。

$$D^{\mathrm{ep}}=\begin{cases}-\dfrac{1+e_0}{\kappa}\sigma & \text{弾性応答（}\mathrm{d}\gamma=0\text{）のとき}\\[2ex] -\dfrac{1+e_0}{\lambda}\sigma_{\mathrm{c}} & \text{弾塑性応答（}\mathrm{d}\gamma>0\text{）のとき}\end{cases} \qquad\cdots\cdots(4.42)$$

4.3 応力計算の方法

3章で述べた変形解析では，サブルーチンcal_stress1において，与えられたひずみεに対して応力σを求める必要がある。線形弾性モデルを用いる場合には，容易に$\sigma=D^{\mathrm{e}}\varepsilon$と計算できる。しかし，前節で導出した弾塑性構成式(4.39)はひずみ増分$\mathrm{d}\varepsilon$と応力増分$\mathrm{d}\sigma$とを関連づける増分形であり，ひずみεから応力σを直接的に計算することはできない。そこで，本章では弾塑性モデルでの応力計算の方法を述べる。

弾塑性モデルの応力計算法には，大きく分けて(i)増分形構成式に基づく方法と，(ii)弾性予測子とリターンマッピングを用いる方法の二つがある[4),5)]。前者では増分形構成式(4.39)について前進差分を考えて応力計算を行うことから，「陽解法」と呼ばれることもある。後者では未知量を含むモデル諸式を陰的に解くことから，「陰解法」とも呼ばれる。ここでは二つの方法の具体的手順を説明するとともに，計算例を示し，各方法の特徴や精度について述べる。

いま，変形解析の第nステップで釣り合い状態が求められており，ひずみ諸量$\{\varepsilon_n, \varepsilon^{\mathrm{p}}_n\}$および応力諸量$\{\sigma_n, \sigma_{\mathrm{c},n}\}$が既知であるとする。第$n$ステップから第$(n+1)$ステップまでのひずみ増分$\Delta\varepsilon$が与えられるとき，(i)，(ii)のいずれかの応力計算法を用いて，第$(n+1)$ステップでのひずみ諸量$\{\varepsilon_{n+1}, \varepsilon^{\mathrm{p}}_{n+1}\}$および応力諸量$\{\sigma_{n+1}, \sigma_{\mathrm{c},n+1}\}$を計算する[注11)]。

4.3.1 増分形構成式に基づく応力計算法

この方法は，増分形で表された弾塑性構成式(4.39)について前進差分を考えることにより，ひずみ増分$\Delta\varepsilon$に対応する応力増分$\Delta\sigma$を計算し，これを既知の応力σ_nに足すことによってσ_{n+1}を求める方法である。

まず，与えられたひずみ増分$\Delta\varepsilon$により，ひずみが次のように更新される。

$$\varepsilon_{n+1}=\varepsilon_n+\Delta\varepsilon \qquad\cdots\cdots(4.43)$$

増分形の弾塑性構成式(4.39)を増分表示すると，

$$\Delta\sigma=D^{\mathrm{ep}}\Delta\varepsilon \qquad\cdots\cdots(4.44)$$

となるが，式(4.42)に示したように，D^{ep}には材料定数e_0, λ, κだけでなく，変形に伴って変化するσやσ_{c}が含まれているため，式(4.44)の計算ではどの時点でのσ, σ_{c}の値を用いるかに注意が必要である。ここでは，第nステップでの既知値$\sigma_n, \sigma_{\mathrm{c},n}$による$D^{\mathrm{ep}}_n$を用いて，

$$\Delta\sigma=D^{\mathrm{ep}}_n\Delta\varepsilon \qquad\cdots\cdots(4.45)$$

とする。その際，式(4.42)に示した弾性・弾塑性の接線係数のどちらを用いるかは，後述のように，直前のステップでの弾性・除荷／塑性負荷の判定結果に基づいて定める。こうして求めた応力増分$\Delta\sigma$を用いて，次のように応力を更新する。

$$\sigma_{n+1}=\sigma_n+\Delta\sigma \qquad\cdots\cdots(4.46)$$

塑性負荷として計算する場合には，次に，式(4.37)を増分表示した式

$$\Delta\gamma=-\frac{D^{\mathrm{e}}_n\Delta\varepsilon}{D^{\mathrm{e}}_n+K^{\mathrm{p}}_n} \qquad\cdots\cdots(4.47)$$

から$\Delta\gamma$を求める。上式でも，第nステップでのD^{e}_nおよびK^{p}_nを用いる。さらに，塑性流れ則(4.33)を増分表示した式

$$\Delta\varepsilon^{\mathrm{p}}=\Delta\gamma\left.\frac{\partial f}{\partial\sigma}\right|_{(\sigma_n,\sigma_{\mathrm{c}},n)}=-\Delta\gamma \quad \left(\because\ \frac{\partial f}{\partial\sigma}=-1\right) \qquad\cdots\cdots(4.48)$$

から塑性ひずみ増分$\Delta\varepsilon^{\mathrm{p}}$が求められるので，弾性ひずみと塑性ひずみをそれぞれ次のように更新する。

$$\varepsilon^{\mathrm{p}}_{n+1}=\varepsilon^{\mathrm{p}}_n+\Delta\varepsilon^{\mathrm{p}},\ \varepsilon^{\mathrm{e}}_{n+1}=\varepsilon_{n+1}-\varepsilon^{\mathrm{p}}_{n+1} \qquad\cdots\cdots(4.49)$$

なお，弾性・除荷として計算する場合は$\Delta\gamma=0$, $\Delta\varepsilon^{\mathrm{p}}=0$であり，式(4.47)，(4.48)の計算は不要である。

塑性負荷の場合には，続いて，式(4.31)の第二式で与えられる増分形の塑性硬化則

$$\Delta\sigma_{\mathrm{c}}=K^{\mathrm{p}}_n\Delta\varepsilon^{\mathrm{p}} \qquad\cdots\cdots(4.50)$$

から計算される$\Delta\sigma_{\mathrm{c}}$を用いて，圧密降伏応力を次のように更新する。

$$\sigma_{\mathrm{c},n+1}=\sigma_{\mathrm{c},n}+\Delta\sigma_{\mathrm{c}}, \qquad\cdots\cdots(4.51)$$

式(4.45)と同様に，式(4.50)でもステップnでのK^{p}_nを用いた。なお，弾性・除荷の場合には式(4.50)，(4.51)の計算は不要であり，$\sigma_{\mathrm{c},n+1}=\sigma_{\mathrm{c},n}$とする。

以上の手続きにより，更新されたひずみと応力の諸量$\{\varepsilon_{n+1}, \varepsilon^{\mathrm{p}}_{n+1}\}$および$\{\sigma_{n+1}, \sigma_{\mathrm{c},n+1}\}$がすべて求められた。最後に，更新された$\sigma_{n+1}$と$\sigma_{\mathrm{c},n+1}$を降伏関数(4.32)に代入して$f_{n+1}=f(\sigma_{n+1}, \sigma_{\mathrm{c},n+1})$の値を求め，このステップ間の応答を次のように判定する。

$$\begin{cases}f_{n+1}\leq 0 \rightarrow \text{弾性・除荷}\\ f_{n+1}>0 \rightarrow \text{塑性負荷}\end{cases} \qquad\cdots\cdots(4.52)$$

この判定結果に基づいて，次ステップの応力計算で式(4.45)中のD^{ep}_nとして式(4.42)の二つの接線係数のどちらを用いるかを定める。

ここで述べた応力計算法では，増分形構成式を用いて前進差分的に応力増分$\Delta\sigma, \Delta\sigma_{\mathrm{c}}$を求めている。そのため，ひずみ増分$\Delta\varepsilon$を大きく設定すると誤差が大きくなる懸念がある。

4.3.2 弾性予測子とリターンマッピングによる応力計算法

弾性予測子とリターンマッピングによる応力計算法を紹介する。前項の方法に比べて計算手続きはやや複雑になるが，大きなひずみ増分を用いた計算でも誤差が発生しにくい等の理由から，最近の弾塑性解析では主にこの方法が用いられている。この応力計算法の特長として，(i)増分形ではない弾性構成式（式(4.30)の第一式）と塑性硬化則（式(4.31)の第一式）を用いること，(ii)弾性予測子と塑性修正子の２段階計算（リターンマッピング）

注11) 前節以前では，変数の前に‘d’を付けて微小増分（微分）を表した。しかし，実際の数値計算では有限の大きさをもつ増分を取り扱うため，本節では変数の前に‘Δ’を付けて増分を表す。

により，厳密に$f=0$（式(4.34)の第三式）を満たす結果が得られることが挙げられる。

まず，前項の方法と同様に，次のようにひずみが更新される。

$$\varepsilon_{n+1}=\varepsilon_n+\Delta\varepsilon \quad\cdots\cdots(4.53)$$

次に，このステップ間で塑性変形が進行せず（$\Delta\varepsilon^{\mathrm{p}}=0$），ステップ間の変形が弾性である（$\Delta\varepsilon^{\mathrm{e}}=\Delta\varepsilon$）と仮定し，次のように試行弾性ひずみを計算する。

$$\varepsilon_{n+1}^{\mathrm{e(tr)}}=\varepsilon_{n+1}-\varepsilon_n^{\mathrm{p}} \quad\cdots\cdots(4.54)$$

上添字の‘(tr)’は試行値（trial value）を意味する。この試行弾性ひずみ$\varepsilon_{n+1}^{\mathrm{e(tr)}}$を弾性構成式（式(4.30)の第一式）に代入して，次のように試行応力を計算する。

$$\sigma_{n+1}^{(\mathrm{tr})}=\sigma_0\exp\left[-\frac{1+e_0}{\kappa}\varepsilon_{n+1}^{\mathrm{e(tr)}}\right] \quad\cdots\cdots(4.55)$$

試行的にステップ間で塑性変形が進行しないと仮定しているので，圧密降伏応力の試行値は前ステップでの値を用いて次のように定められる。

$$\sigma_{\mathrm{c},n+1}^{(\mathrm{tr})}=\sigma_{\mathrm{c},n} \quad\cdots\cdots(4.56)$$

式(4.54)，(4.55)，(4.56)でそれぞれ定められる試行値$\varepsilon_{n+1}^{\mathrm{e(tr)}}$, $\sigma_{n+1}^{(\mathrm{tr})}$, $\sigma_{\mathrm{c},n+1}^{(\mathrm{tr})}$をまとめて弾性予測子と呼ぶ。

続いて，試行値$\sigma_{n+1}^{(\mathrm{tr})}$, $\sigma_{\mathrm{c},n+1}^{(\mathrm{tr})}$を降伏関数(4.32)に代入して

$$f_{n+1}^{(\mathrm{tr})}=f(\sigma_{n+1}^{(\mathrm{tr})},\sigma_{\mathrm{c},n+1}^{(\mathrm{tr})})=\sigma_{\mathrm{c},n+1}^{(\mathrm{tr})}-\sigma_{n+1}^{(\mathrm{tr})} \quad\cdots\cdots(4.57)$$

とし，その値により次のように負荷判定を行う。

$$\begin{cases} f_{n+1}^{(\mathrm{tr})}\leq 0 \rightarrow \text{弾性・除荷} \\ f_{n+1}^{(\mathrm{tr})}>0 \rightarrow \text{塑性負荷} \end{cases} \quad\cdots\cdots(4.58)$$

式(4.58)で弾性・除荷と判定された場合には，次式のように試行値を第（$n+1$）ステップでの値として採用する。

$$\varepsilon_{n+1}^{\mathrm{e}}=\varepsilon_{n+1}^{\mathrm{e(tr)}},\ \sigma_{n+1}=\sigma_{n+1}^{(\mathrm{tr})},\ \sigma_{\mathrm{c},n+1}=\sigma_{\mathrm{c},n+1}^{(\mathrm{tr})} \quad\cdots\cdots(4.59)$$

また，この場合は言うまでもなく$\Delta\gamma=0$である。あとは$\varepsilon_{n+1}^{\mathrm{p}}=\varepsilon_{n+1}-\varepsilon_{n+1}^{\mathrm{e}}$とすることにより，更新された諸量$\{\varepsilon_{n+1},\varepsilon_{n+1}^{\mathrm{p}}\}$および$\{\sigma_{n+1},\sigma_{\mathrm{c},n+1}\}$が求められる。

一方，式(4.58)で塑性負荷と判定された場合には，$\Delta\gamma>0$であり，式(4.34)の第三式，すなわち

$$f_{n+1}=f(\sigma_{n+1},\sigma_{\mathrm{c},n+1})=\sigma_{\mathrm{c},n+1}-\sigma_{n+1}=0 \quad\cdots\cdots(4.60)$$

を満たす諸量を以下に述べる手順で求める。

まず，塑性流れ則(4.33)を次のように増分表示する。

$$\Delta\varepsilon^{\mathrm{p}}=\Delta\gamma\left.\frac{\partial f}{\partial\sigma}\right|_{(\sigma_{n+1},\sigma_{\mathrm{c},n+1})}=-\Delta\gamma \quad \left(\because\ \frac{\partial f}{\partial\sigma}=-1\right) \quad\cdots\cdots(4.61)$$

ここでは後退オイラー差分を用いるため，陽解法の式(4.48)とは異なり，上式では時刻t_{n+1}での$\dfrac{\partial f}{\partial\sigma}$を用いていることに着目されたい（ただし，式(4.32)の降伏関数では常に$\dfrac{\partial f}{\partial\sigma}=1$となるため，陽解法と陰解法との差異はない）。試行弾性ひずみの計算式(4.54)および$\varepsilon_{n+1}^{\mathrm{p}}=\varepsilon_n^{\mathrm{p}}+\Delta\varepsilon^{\mathrm{p}}$の関係より，ひずみの加算分解式(4.29)は次のように書き換えられる。

$$\begin{aligned}\varepsilon_{n+1}^{\mathrm{e}}&=\varepsilon_{n+1}-\varepsilon_{n+1}^{\mathrm{p}}=\varepsilon_{n+1}-(\varepsilon_n^{\mathrm{p}}-\Delta\gamma)\\&=\varepsilon_{n+1}^{\mathrm{e(tr)}}+\Delta\gamma\end{aligned} \quad\cdots\cdots(4.62)$$

式(4.54)から分かるように，$\varepsilon_{n+1}^{\mathrm{e(tr)}}$は既知量$\varepsilon_{n+1}$, $\varepsilon_n^{\mathrm{p}}$から計算されるので，これも既知量である。式(4.60)，(4.62)，および弾性構成式(4.30)と塑性硬化則(4.31)の四つの式を連立させて，四つの未知数$\varepsilon_{n+1}^{\mathrm{e}}$, σ_{n+1}, $\sigma_{\mathrm{c},n+1}$, $\Delta\gamma$について解くことができる。解くべき方程式を整理すると，

$$\begin{cases} y_1=\varepsilon_{n+1}^{\mathrm{e}}-\varepsilon_{n+1}^{\mathrm{e(tr)}}-\Delta\gamma=0 \\ y_2=\sigma_{n+1}-\sigma_0\exp\left[-\dfrac{1+e_0}{\kappa}\varepsilon_{n+1}^{\mathrm{e}}\right]=0 \\ y_3=\sigma_{\mathrm{c},n+1}-\sigma_{\mathrm{c}0}\exp\left[-\dfrac{1+e_0}{\lambda-\kappa}(\varepsilon_{n+1}-\varepsilon_{n+1}^{\mathrm{e}})\right]=0 \\ y_4=\sigma_{\mathrm{c},n+1}-\sigma_{n+1}=0 \end{cases} \quad\cdots\cdots(4.63)$$

となる。ここでは各式の値を$y_1,\cdots,y_4$とおいており，これらを成分とするベクトル$\boldsymbol{y}=(y_1,y_2,y_3,y_4)^{\mathrm{T}}$および未知数ベクトル$\boldsymbol{x}=(\varepsilon_{n+1}^{\mathrm{e}},\sigma_{n+1},\sigma_{\mathrm{c},n+1},\Delta\gamma)^{\mathrm{T}}$を定義すれば，式(4.63)は形式的に

$$\boldsymbol{y}(\boldsymbol{x})=0 \quad\cdots\cdots(4.64)$$

と書くことができる。ここで，$\boldsymbol{y}$は$\boldsymbol{x}$に関する非線形方程式になっているので，ニュートン・ラプソン法（以下ではNR法と呼ぶ）を用いて解く。変数に下添字（k）を付けて第k回目の反復での値を表すこととし，式(4.64)を$\boldsymbol{x}$の修正量$\boldsymbol{\delta x}$について線形化すると，

$$\boldsymbol{y}(\boldsymbol{x}_{(k)})+\frac{\partial\boldsymbol{y}(\boldsymbol{x}_{(k)})}{\partial\boldsymbol{x}}\cdot\boldsymbol{\delta x}=\boldsymbol{0} \quad\cdots\cdots(4.65)$$

となる。ここで，$\dfrac{\partial\boldsymbol{y}(\boldsymbol{x}_{(k)})}{\partial\boldsymbol{x}}$はヤコビ行列であり，ここでは4×4行列となる。その導出は容易であるため，具体形は省略する。式(4.65)を$\boldsymbol{\delta x}$について解くと，

$$\partial\boldsymbol{x}=-\left(\frac{\partial\boldsymbol{y}(\boldsymbol{x}_{(k)})}{\partial\boldsymbol{x}}\right)^{-1}\cdot\boldsymbol{y}(\boldsymbol{x}_{(k)}) \quad\cdots\cdots(4.66)$$

となり，修正量$\boldsymbol{\delta x}$を用いて$\boldsymbol{x}$を次のように更新する。

$$\boldsymbol{x}_{(k+1)}=\boldsymbol{x}_{(k)}+\boldsymbol{\delta x} \quad\cdots\cdots(4.67)$$

更新値$\boldsymbol{x}_{(k+1)}$を式(4.64)に代入し，次式で収束を確認する。

$$\|\boldsymbol{y}(\boldsymbol{x}_{(k+1)})\|<\mathrm{TOL} \quad\cdots\cdots(4.68)$$

ここで，$\|\boldsymbol{y}\|=\sqrt{(y_1)^2+\cdots+(y_4)^2}$はベクトル$\boldsymbol{y}$のノルム（長さ）である。また，TOLは収束判定基準（tolerance）であり，一般に1.0×10^{-6}～10^{-12}程度の値が用いられる。収束条件(4.68)が満たされていれば反復を終了する。収束していなければ，$\boldsymbol{x}_{(k+1)}$を$\boldsymbol{x}_{(k)}$に置き換えて式(4.66)から反復計算を続ける。

以上が一般的なリターンマッピングの手続きであるが，ここで考えている単純な構成モデルでは，上述のように非線形連立方程式を扱わずに，以下のように単一式を解くだけで済む。連立方程式(4.63)の第一式を第二式に代入する。また，$\varepsilon_{n+1}^{\mathrm{p}}=\varepsilon_n^{\mathrm{p}}+\Delta\varepsilon^{\mathrm{p}}$の関係と式(4.61)を式(4.63)の第三式に代入すると，以下の2式を得る。

$$\sigma_{n+1}=\sigma_0\exp\left[-\frac{1+e_0}{\kappa}(\varepsilon_{n+1}^{\mathrm{e(tr)}}+\Delta\gamma)\right], \quad\cdots\cdots(4.69)$$

$$\sigma_{\mathrm{c},n+1}=\sigma_{\mathrm{c}0}\exp\left[-\frac{1+e_0}{\lambda-\kappa}(\varepsilon_n^{\mathrm{p}}-\Delta\gamma)\right] \quad\cdots\cdots(4.70)$$

上の２式を式(4.63)の第四式に代入すると，

$$\sigma_{c0}\exp\left[-\frac{1+e_0}{\lambda-\kappa}(\varepsilon_n^{p}-\Delta\gamma)\right]$$

$$-\sigma_0\exp\left[-\frac{1+e_0}{\kappa}(\varepsilon_{n+1}^{e(tr)}+\Delta\gamma)\right]=0 \quad\cdots\cdots\cdots(4.71)$$

となり，未知数は$\Delta\gamma$だけである。ただし，これは$\Delta\gamma$に関する非線形方程式なので，NR 法を使って解く。そこで，上式を

$$F(\Delta\gamma)=0 \quad\cdots\cdots\cdots(4.72)$$

とおいて，$\Delta\gamma$の修正量$\delta(\Delta\gamma)$について線形化すると，

$$F(\Delta\gamma_{(k)})+F'(\Delta\gamma_{(k)})\delta(\Delta\gamma)=0 \quad\cdots\cdots\cdots(4.73)$$

となる。ここで，式(4.69)，(4.70)，および微分のチェインルールより，$F'(\Delta\gamma_{(k)})$の具体形は次のとおりである。

$$F'(\Delta\gamma)=\sigma_{c,\,n+1}\frac{1+e_0}{\lambda-\kappa}-\sigma_{n+1}\left(-\frac{1+e_0}{\kappa}\right) \quad\cdots\cdots\cdots(4.74)$$

あとは，式(4.73)を$\delta(\Delta\gamma)$について解き，次のように$\Delta\gamma$を更新する。

$$\delta(\Delta\gamma)=-\frac{F(\Delta\gamma_{(k)})}{F'(\Delta\gamma_{(k)})},\ \Delta\gamma_{(k+1)}=\Delta\gamma_{(k)}+\delta(\Delta\gamma) \quad\cdots\cdots\cdots(4.75)$$

更新値$\Delta\gamma_{(k+1)}$を式(4.72)に代入し，判定式$|F(\Delta\gamma_{(k+1)})|<$TOL により収束を確認する。こうして求められた$\Delta\gamma$を式(4.62)，(4.69)，(4.70)に代入することにより，更新された諸量$\{\varepsilon_{n+1},\varepsilon_{n+1}^{p}\}$および$\{\sigma_{n+1},\sigma_{c,\,n+1}\}$がすべて求められる。

有限要素解析で要素接線剛性を計算する際に用いる接線係数について述べておく。前項の陽解法では式(4.42)に示した増分形構成式の接線係数D^{ep}を用いればよかった。しかし，陰解法の応力計算を用いる場合，非線形釣り合い式を NR 法で解く際に良好な収束を得るためには，応力更新アルゴリズムに整合した接線係数$\bar{D}^{ep}$を用いる必要がある。これをアルゴリズム接線係数と呼ぶ[注12)]。アルゴリズム接線係数は次のように定義される。

$$\bar{D}_{n+1}^{ep}=\frac{\partial\sigma_{n+1}}{\partial\varepsilon_{n+1}} \quad\cdots\cdots\cdots(4.76)$$

以下では，アルゴリズム接線係数の具体形を導出する。

弾性・除荷の場合は，式(4.54)，(4.59)より$d\varepsilon_{n+1}=d\varepsilon_{n+1}^{e}$であるから，式(4.76)の微分演算は式(4.17)と同様であり，$\bar{D}_{n+1}^{ep}=D_{n+1}^{e}$となる。

塑性負荷の場合のアルゴリズム接線係数は，以下のようにして導出される。式(4.54)，(4.69)より，弾性構成式は

$$\sigma_{n+1}=\sigma_0\exp\left[-\frac{1+e_0}{\kappa}(\varepsilon_{n+1}-\varepsilon_n^{p}+\Delta\gamma)\right] \quad\cdots\cdots(4.77)$$

と表すことができる。ε_n^{p}は既知量であるが，$\Delta\gamma$は応力計算において未知量であることに注意して，式(4.77)を更新時のひずみε_{n+1}で微分すると，

$$\bar{D}_{n+1}^{ep}=\frac{\partial\sigma_{n+1}}{\partial\varepsilon_{n+1}}=\underbrace{-\frac{1+e_0}{\kappa}\sigma_{n+1}}_{=D_{n+1}^{e}}\left(1+\frac{\partial(\Delta\gamma)}{\partial\varepsilon_{n+1}}\right) \quad\cdots\cdots(4.78)$$

となり，式中の$\frac{\partial(\Delta\gamma)}{\partial\varepsilon_{n+1}}$を求める必要がある。塑性適合条件(4.35)から弾塑性応答時は$df=0$であることを考慮すると，

$$\frac{\partial f_{n+1}}{\partial\varepsilon_{n+1}}=0 \quad\cdots\cdots\cdots(4.79)$$

が成り立つので，式(4.71)の左辺をε_{n+1}で微分すると，

$$\frac{\partial f_{n+1}}{\partial\varepsilon_{n+1}}=\frac{1+e_0}{\lambda-\kappa}\sigma_{c,\,n+1}\frac{\partial(\Delta\gamma)}{\partial\varepsilon_{n+1}}$$

$$-\left(-\frac{1+e_0}{\kappa}\right)\sigma_{n+1}\left(1+\frac{\partial(\Delta\gamma)}{\partial\varepsilon_{n+1}}\right)=0 \quad\cdots(4.80)$$

となる。これを$\frac{\partial(\Delta\gamma)}{\partial\varepsilon_{n+1}}$について解き，塑性負荷時において$\sigma_{n+1}=\sigma_{c,\,n+1}$であることを考慮すると，

$$\frac{\partial(\Delta\gamma)}{\partial\varepsilon_{n+1}}=\frac{-\dfrac{1+e_0}{\kappa}\sigma_{c,\,n+1}}{\dfrac{1+e_0}{\lambda-\kappa}\sigma_{c,\,n+1}+\dfrac{1+e_0}{\kappa}\sigma_{c,\,n+1}}=-\frac{\lambda-\kappa}{\lambda} \quad\cdots(4.81)$$

となる。これを式(4.78)へ代入し，$\sigma_{n+1}=\sigma_{c,\,n+1}$とおくと，塑性負荷時のアルゴリズム接線係数

$$\bar{D}_{n+1}^{ep}=-\frac{1+e_0}{\kappa}\sigma_{c,\,n+1}\left(1-\frac{\lambda-\kappa}{\lambda}\right)=-\frac{1+e_0}{\lambda}\sigma_{c,\,n+1} \quad\cdots\cdots\cdots(4.82)$$

が得られる[注13)]。

4.3.3　プログラムへの実装

弾塑性モデルの応力計算法を実際にプログラムに実装する。本講座では陰解法による有限要素法を扱っているので，**4.3.1**で述べた増分形構成式に基づく応力計算法については割愛し，**4.3.2**の弾性予測子とリターンマッピングによる応力計算法の実装のみを示す。有限要素法による変形解析の手続き全体のうち，弾塑性の応力計算は３章で示したフローチャート（**図—3.4**）中の「応力計算」に相当する。プログラムへの実装では，線形弾性モデルのサブルーチン cal_stress1 を弾塑性モデルのものに入れ換えればよい。

4.3.4　計算例

これまで紹介した二つの応力計算法の精度を比較する。

リスト—4.1　応力の計算（弾性・弾塑性の場合分け）

行　ソースコード

```
SUBROUTINE cal_stress
  USE common
  IMPLICIT NONE
  SELECT CASE(type_estr_m(mat_e(ielem)))
    CASE(2)
      CALL cal_stress2
    CASE(3)
```

注12)　整合接線係数，コンシステント接線係数などと呼ばれることもある。増分形の応力−ひずみ関係(4.39)における弾塑性接線係数D^{ep}（これは一般に「連続体接線係数」などと呼ばれる）と区別するために，ここでは上線を付けて$\bar{D}^{ep}$と表記している。

注13)　このアルゴリズム接線係数は，式(4.42)に示した増分形構成式の弾塑性接線係数D_{ep}と一致している。しかし，これは一次元の弾塑性モデルだけに言えることであり，一般的な三次元モデルでは，アルゴリズム接線係数と弾塑性接線係数とは必ずしも一致しない。

行	ソースコード
8	`      CALL cal_stress3`
9	`  END SELECT`
10	`END SUBROUTINE cal_stress`

行	説明文
4	要素番号 ielem での材料番号 mat_e とその材料の種類 type_estr_m によって場合分け。
5	type_estr_m=2 ならば弾性モデル。
6	弾性モデルの応力計算サブルーチン cal_stress2 を呼び出す。
7	type_estr_m=3 ならば弾塑性モデル。
8	弾塑性モデルの応力計算サブルーチン cal_stress3 を呼び出す。

リスト—4.2　応力の計算（弾性の場合）

行	ソースコード
1	`SUBROUTINE cal_stress2`
2	`  USE common`
3	`  IMPLICIT NONE`
4	`  REAL(p):: n_0s,voidr_0,sts0,kappa`
5	`  n_0s=prop_base_m(mat_e(ielem),1)`
6	`  sts0=prop_estr_m(mat_e(ielem),1)`
7	`  kappa=prop_estr_m(mat_e(ielem),2)`
8	`  voidr_0=n_0s/(1.0_p-n_0s)`
9	`  sts_i=sts0*EXP(-((1.0_p+voidr_0)/kappa)*stn_i)`
10	`  sts_e_i(ielem,iintp)=sts_i`
11	`  dep_i=-((1.0_p+voidr_0)/kappa)*sts_i`
12	`END SUBROUTINE cal_stress2`

行	説明文
4	ローカル変数の定義。
5	初期間隙率 n_0s に入力値を代入。
6	基準応力 sts0 に入力値を代入。
7	弾性圧縮指数（膨潤指数）kappa に入力値を代入。
8	初期間隙率 n_0s から初期間隙比 voidr_0 を計算。
9	ひずみ stn_i を弾性構成式(4.16)に代入して応力 sts_i を計算。
10	9行目で計算した応力 sts_i を要素番号 ielem，積分点 iintp での更新された応力 sts_e_i として代入。
11	応力 sts_i を式(4.18)に代入して弾性接線係数を計算し，これをアルゴリズム接線係数 dep_i に代入。

リスト—4.3　応力の計算（弾塑性の場合）

行	ソースコード
1	`SUBROUTINE cal_stress3`
2	`  USE common`
3	`  IMPLICIT NONE`
4	`  INTEGER:: id_ep,iter_retmap`
5	`  REAL(p):: n_0s,voidr_0,sts0,iso_hp0,kappa,lamda`
6	`  REAL(p):: iso_hpt,iso_hp_trial,stn_elast_trial`
7	`  REAL(p):: stnt_plast,stn_elast,stn_plast,sts_trial`
8	`  REAL(p):: fyield,del_fyield, &` `            del_gamma,correct_del_gamma`
9	`  n_0s=prop_base_m(mat_e(ielem),1)`
10	`  sts0=prop_estr_m(mat_e(ielem),1)`
11	`  kappa=prop_estr_m(mat_e(ielem),2)`
12	`  iso_hp0=prop_estr_m(mat_e(ielem),3)`
13	`  lamda=prop_estr_m(mat_e(ielem),4)`
14	`  voidr_0=n_0s/(1.0_p-n_0s)`
15	`  iso_hpt=iso_hpt_e_i(ielem,iintp)`
16	`  stnt_plast=-((lamda-kappa)/(1.0_p+voidr_0)) &` `                *LOG(iso_hpt/iso_hp0)`
17	`  stn_elast_trial=stn_i-stnt_plast`
18	`  sts_trial=sts0 &` `    *EXP(-((1.0_p+voidr_0)/kappa)*stn_elast_trial)`
19	`  iso_hp_trial=iso_hpt`
20	`  fyield=iso_hp_trial-sts_trial`
21	`  IF(fyield <= 0.0_p) THEN`
22	`    sts_i=sts_trial`
23	`    iso_hp_i=iso_hp_trial`
24	`    dep_i=-((1.0_p+voidr_0)/kappa)*sts_i`
25	`  ELSEIF(fyield > 0.0_p) THEN`
26	`    del_gamma=0.0_p`
27	`    iteration_retmap: DO iter_retmap=1,max_iter`
28	`      stn_elast=stn_elast_trial+del_gamma`
29	`      stn_plast=stn_i-stn_elast`
30	`      sts_i=sts0 &` `        *EXP(-((1.0_p+voidr_0)/kappa)*stn_elast)`

31	`      iso_hp_i=iso_hp0 &`
	`        *EXP(-((1.0_p+voidr_0)/(lamda-kappa)) &`
	`             *stn_plast)`

32	`      fyield=iso_hp_i-sts_i`
33	`      IF(ABS(fyield) < er_tol) EXIT`
34	`      IF(iter_retmap == max_iter) STOP`

35	`      del_fyield=iso_hp_i &`
	`           *((1.0_p+voidr_0)/(lamda-kappa)) &`
	`             -sts_i*(-(1.0_p+voidr_0)/kappa)`

36	`      correct_del_gamma=-fyield/del_fyield`
37	`      del_gamma=del_gamma+correct_del_gamma`
38	`    END DO iteration_retmap`
39	`    dep_i=-((1.0_p+voidr_0)/lamda)*iso_hp_i`
40	`  ENDIF`
41	`  sts_e_i(ielem,iintp)=sts_i`
42	`  iso_hp_e_i(ielem,iintp)=iso_hp_i`
43	`END SUBROUTINE cal_stress3`

行	説明文
4	ローカル変数の定義。8行目まで。
9	初期間隙率 n_0s に入力値を代入。
10	基準応力 sts0 に入力値を代入。
11	弾性圧縮指数（膨潤指数）kappa に入力値を代入。
12	基準状態の圧密降伏応力 iso_hp0 に入力値を代入。
13	弾塑性圧縮指数 lamda に入力値を代入。
14	初期間隙率 n_0s から初期間隙比 voidr_0 を計算。
15	要素番号 ielem，積分点 iintp の直前のステップでの圧密降伏応力を iso_hpt に代入。
16	式(4.19)に iso_hpt を代入して，直前のステップでの塑性ひずみ stnt_plast を計算。
17	式(4.54)に基づき，現ステップでのひずみ stn_i から stnt_plast を差し引くことにより試行弾性ひずみ stn_elast_trial を計算。
18	試行弾性ひずみ stn_elast_trial を式(4.55)に代入し，試行応力 sts_trial を計算。
19	式(4.56)に基づいて圧密降伏応力の試行値 iso_hp_trial を計算。

20	試行応力`sts_trial`と圧密降伏応力の試行値`iso_hp_trial`を式(4.57)に代入し，降伏関数の値`fyield`を計算。
21	式(4.58)に基づいて負荷判定を行う。弾性・除荷と判定されたら，22～24行目の手続きを行う。
22	試行応力`sts_trial`をそのまま更新値`sts_i`として採用。式(4.59)の第二式に相当。
23	圧密降伏応力の試行値`iso_hp_trial`をそのまま更新値`iso_hp_i`として採用。式(4.59)の第三式に相当。
24	弾性・除荷の場合のアルゴリズム接線係数`dep_i`を計算。式(4.18)に相当。
25	式(4.58)に基づいて負荷判定を行う。塑性負荷と判定されたら，26～39行目の手続き（リターンマッピング）を行う。
26	増分的な塑性乗数`del_gamma`の初期値としてゼロを代入。
27	式(4.72)をNR法で解く反復計算を開始。`iteration_retmap`のDOループ。収束するまで，最大`max_iter`回の反復計算を行う。
28	式(4.62)より，弾性ひずみの暫定更新値`stn_elast`を計算。
29	全ひずみから弾性ひずみを差し引くことにより，塑性ひずみの暫定更新値`stn_plast`を計算。
30	弾性構成式(4.30)に弾性ひずみ`stn_elast`を代入し，応力の暫定更新値`sts_i`を計算。
31	塑性硬化則(4.31)に塑性ひずみ`stn_plast`を代入し，圧密降伏応力の暫定更新値`iso_hp_i`を計算。
32	暫定更新値`sts_i`と`iso_hp_i`を降伏関数(4.60)に代入して`fyield`値を計算。
33	`fyield`の絶対値が収束判定値`er_tol`未満となったら，反復計算が収束したと判断し，DOループから抜ける。
34	反復回数`iter_retmap`が上限値`max_iter`に達しても収束しなかったら，計算を中止。
35	式(4.74)の導関数$F'(\varDelta\gamma)$の値`del_fyield`を計算。
36	式(4.75)の第一式より，塑性乗数の修正量`correct_del_gamma`を計算。
37	式(4.75)の第二式のように塑性乗数`del_gamma`を更新する。
38	`iteration_retmap`のDOループ末尾。28行目へ。
39	塑性負荷の場合のアルゴリズム接線係数`dep_i`を計算。式(4.82)に相当。
40	21行目と25行目の場合分けが終了。
41	収束計算で求めた応力`sts_i`を要素番号`ielem`，積分点`iintp`での更新された応力`sts_e_i`として代入。
42	同様に，圧密降伏応力`iso_hp_i`を`iso_hp_e_i`に代入。

基準状態を（σ_0, e_0）=（−10.0 kPa, 1.80），そのときの圧密降伏応力を$\sigma_{c0}=-200$ kPaとし，圧縮指数を$\lambda=0.130$，膨潤指数を$\kappa=0.018$とした。増分ステップ数を$N_{\mathrm{step}}=5$, 10, 50, 1000として，最終的にひずみが$\varepsilon=-0.10$（10%の圧縮ひずみ）に達するように等分割して$\varDelta\varepsilon$を与えた。以下では，**4.3.1**の応力計算法を「計算法1」，**4.3.2**の応力計算法を「計算法2」と呼ぶ。

二つの計算法で求めたe-ln σ関係を**図―4.3**(a)，(b)に示す。ここでは式(4.10)を用いてひずみεから間隙比eを求めた。大きなひずみ増分で計算を行うと「計算法1」では計算精度が低下しているが，「計算法2」では良好な精度を確保している。これらの結果から，「計算法1」を用いる際にはひずみ増分の設定に注意を払う必要がある。一方，「計算法2」では大きなひずみ増分を用いて効率的な解析が可能である。最後に，「計算法2」での式(4.72)の反復計算における収束の様子を**図―4.4**に示す。各計算ケースの最終ステップにおける収束状況を示した。収束判定基準は$|F(\varDelta\gamma)|<1.0\times10^{-10}$とした。どのケースにおいてもNR法本来の二次収束を示している。このような収束状況の確認は，「計算法2」のプログラム作成の際，正しいプログラムになっているか否かのチェックの一つとして有効である。



(a) 計算法1

(b) 計算法2

図―4.3 二つの方法により求めたe-ln σ関係



図―4.4 「計算法2」での収束計算の様子

参考文献

1) 土の構成モデルと数値地盤解析への応用（小特集），土と基礎，Vol. 52, No. 8, 2004.

2) 木村　孟ほか：講座「カムクレイに学ぶ」，土と基礎，Vol. 41, No. 5, 1993からVol. 42, No. 4, 1994にかけて全12回連載.

3) 地盤工学会編：土の弾塑性構成モデル（地盤工学・基礎理論シリーズ3），2009.

4) 寺田賢二郎監訳：塑性の計算力学―理論と応用―（仮題），森北出版（発刊予定）.

5) Simo, J. C. and Hughes, T. J. R.: Computational Inelasticity, Springer–Verlag, 1998.